

Extra examples

1 – The HC-SR505 PIR sensor

The HC-SRS505 PIR sensor that can be found in the lab kits is an effective way of detecting motion. It is based on infrared technology and has both high sensitivity and high reliability. It is widely used in lots of automation applications, particularly in battery-powered applications where it's small size and low-power operation mode is particularly useful.

One of the key things to note here is that the sensor must be powered from the 5V supply on the Arduino (it's operating range is 4.5 – 20V). If you use the normal VCC pin on the Arduino MKR WiFi 1010 it will not supply enough voltage and motion detection will be very unreliable! Figure 1 shows a circuit schematic for using the PIR sensor.




Figure 1 – PIR sensor circuit

(Note that the PIR sensor we have in our kits looks a little different, but the pin outs are the same!).

We can use this PIR sensor as a simple digital input. The only thing to not here is that, when motion is detected, it latches HIGH for 8s +/- 30% (this is specified in the datasheet). Code listing 1 shows an example motion detector application.

Code Listing 1: A motion detector

```
/**
 * 1_hc_sr505_pir_sensor.ino
 *
 * simple sketch to show the operation of the HC-SR505 pir sensor
 *
 * note the component info can be found at:
 * https://www.elecrow.com/wiki/index.php?title=HC-SR505\_Mini\_PIR\_Motion\_Sensor
 *
 * key points are:
 * 1/ the delay time (before motion is no longer detected) is ~11s
 * 2/ the sensor needs to be powered from 5v or motion detection will
 *    be really unreliable
 *
 * author: alex shenfield
 * date: 30/01/2023
 */

// SENSORS AND OUTPUTS

// set the led and pir sensor pins
const int led_pin = 6;
const int pir_pin = 7;

// TIMING VARIABLES

// check every 500 ms (unless motion is detected!)
long previous_time = 0;
long check_interval = 500;

// CODE

// this method runs once (when the sketch starts)
void setup()
{
    // set up serial comms for debugging
    Serial.begin(115200);
    while (!Serial);
    Serial.println("starting security detection system ...");

    // set the pir sensor pin as an input and initialise it low
    pinMode(pir_pin, INPUT);
    digitalWrite(pir_pin, LOW);

    // set the led pin as an output and initialise it low
    pinMode(led_pin, OUTPUT);
    digitalWrite(led_pin, LOW);
}
```

```
// this methods loops continuously
void loop()
{
    // check the pir sensor every 11s
    unsigned long current_time = millis();
    if (current_time - previous_time > check_interval)
    {
        previous_time = current_time;

        // if the pir sensor reads high (i.e. motion is detected) ...
        if (digitalRead(pir_pin) == HIGH)
        {
            // print an alert message and turn the led on
            Serial.println("motion detected");
            digitalWrite(led_pin, HIGH);

            // the pir datasheet states that it's delay time is 8s +/- 30%,
            // so, if we detect motion, there's no point in checking again
            // until ~11s has elapsed
            check_interval = 11000;
        }
        else
        {
            // otherwise, no motion is detected so turn led off ...
            Serial.println("no motion detected");
            digitalWrite(led_pin, LOW);

            // ... and check again in 500 ms!
            check_interval = 500;
        }
    }
}
```